Introducing Stata

This chapter will run through a sample work session, introducing you to a few of the basic tasks that can be done in Stata, such as opening a dataset, investigating the contents of the dataset, using some descriptive statistics, making some graphs, and doing a simple regression analysis. As you would expect, we will only brush the surface of many of these topics. This approach should give you a sample of what Stata can do and how Stata works. There will be brief explanations along the way, with references to chapters later in this book as well as to the system help and other Stata manuals. We will run through the session by using both menus and dialogs and Stata’s commands so that you can become familiar with them both. If you see that your menus and dialogs are not in English, we recommend that you (temporarily) change the locale used by Stata to English, so that you can work along with the examples. See [P] set locale_ui for how to do this.

If you are using Stata(console) rather than Stata(GUI) under Unix, you will not be able to use the menus and dialogs as they are mentioned here. However, the commands used to produce the output are included in the output in each case, so you can still work through the examples.

Take a seat at your computer, put on some good music, and work along with the book.

Sample session

The dataset that we will use for this session is a set of data about vintage 1978 automobiles sold in the United States.

To follow along by pointing and clicking, note that the menu items are given by Menu > Menu item > Submenu item > etc. To follow along by using the Command window, type the commands that follow a dot (.) in the boxed listings below into the small window at the bottom of the Stata interface. When there is something to note about the structure of a command, it will be pointed out as a “Syntax note”.

Start by loading the auto dataset, which is included with Stata. To use the menus,
1. Select File > Example datasets....
2. Click on Example datasets installed with Stata.
3. Click on use for auto.dta.
The result of this command is fourfold:
- The following output appears in the large Results window:

```
.sysuse auto
(1978 Automobile Data)
```

The output consists of a command and its result. The command, sysuse auto.dta, is bold and follows the dot (.). The result, (1978 Automobile Data), is in the standard face here and is a brief description of the dataset.

Note: If a command intrigues you, you can type help commandname in the Command window to find help. If you want to explore at any time, Help > Search... can be informative.
• The same command, `sysuse auto.dta`, appears in the tall History window to the left. The History window keeps track of all commands Stata has run, successful and unsuccessful. The commands can then easily be rerun. See [GSU 2 The Stata user interface] for more information.
• A series of variables appears in the small Variables window to the upper right.
• Some information about `make`, the first variable in the dataset, appears in the small Properties window to the lower right.

You could have opened the dataset by typing `sysuse auto` in the Command window and pressing `Enter`. Try this now. `sysuse` is a command that loads (uses) example (system) datasets. As you will see during this session, Stata commands are often simple enough that it is faster to use them directly. This will be especially true once you become familiar with the commands you use the most in your daily use of Stata.

Syntax note: In the above example, `sysuse` is the Stata command, whereas `auto` is the name of a Stata data file.

Simple data management

We can get a quick glimpse at the data by browsing them in the Data Editor. This can be done by clicking on the Data Editor (Browse) button, 📊, or by selecting Data > Data Editor > Data Editor (Browse) from the menus or by typing the command `browse`.

No command is issued when clicking on either Data Editor button because opening the Data Editor has no effect on the dataset or any possible analysis.

When the Data Editor window opens, you can see that Stata regards the data as one rectangular table. This is true for all Stata datasets. The columns represent variables, whereas the rows represent observations. The variables have somewhat descriptive names, whereas the observations are numbered.

The data are displayed in multiple colors—at first glance, it appears that the variables listed in black are numeric, whereas those that are in colors are text. This is worth investigating. Click on a cell under the `make` variable: the input box at the top displays the make of the car. Scroll to the right until you see the `foreign` variable. Click on one of its cells. Although the cell may display “Domestic”, the input box displays a 0. This shows that Stata can store categorical data as numbers but display
human-readable text. This is done by what Stata calls value labels. Finally, under the rep78 variable, which looks to be numeric, there are some cells containing just a dot (.) . The dots correspond to missing values.

Looking at the data in this fashion, though comfortable, lends little information about the dataset. It would be useful for us to get more details about what the data are and how the data are stored. Close the Data Editor by clicking on its close button.

We can see the structure of the dataset by describing its contents. This can be done either by going to Data > Describe data > Describe data in memory or in a file in the menus and clicking on OK or by typing describe in the Command window and pressing Enter. Regardless of which method you choose, you will get the same result:

```
. describe
Contains data from /usr/local/stata16/ado/base/a/auto.dta
obs: 74 1978 Automobile Data
vars: 12 13 Apr 2018 17:45
(_dta has notes)

variable name  type    display format label               variable label
make           str18   %-18s          Make and Model
price          int     %8.0gc        Price
mpg            int     %8.0g         Mileage (mpg)
rep78          int     %8.0g         Repair Record 1978
headroom       float   %6.1f         Headroom (in.)
trunk          int     %8.0g         Trunk space (cu. ft.)
weight         int     %8.0gc        Weight (lbs.)
length         int     %8.0g         Length (in.)
turn           int     %8.0g         Turn Circle (ft.)
displacement   int     %8.0g         Displacement (cu. in.)
gear_ratio     float   %6.2f        Gear Ratio
foreign        byte    %8.0g         origin Car type

Sorted by: foreign
```

At the top of the listing, some information is given about the dataset, such as where it is stored on disk and when the dataset was last saved. The bold 1978 Automobile Data is the short description that appeared when the dataset was opened and is referred to as a data label by Stata. The phrase _dta has notes informs us that there are notes attached to the dataset. We can see what notes there are by typing notes in the Command window:

```
. notes
_dta:
   1. from Consumer Reports with permission
```

Here we see a short note about the source of the data.

Looking back at the listing from describe, we can see that Stata keeps track of more than just the raw data. Each variable has the following:

- A variable name, which is what you call the variable when communicating with Stata. Variable names are one type of Stata name. See [U] 11.3 Naming conventions.
- A *storage type*, which is the way Stata stores its data. For our purposes, it is enough to know that types like, say, `str#` are *string*, or text, variables, whereas all others in this dataset are numeric. While there are none in this dataset, Stata also allows arbitrarily long strings, or `strL`s. `strL`s can also contain binary information. See [U] 12.4 Strings.

- A *display format*, which controls how Stata displays the data in tables. See [U] 12.5 Formats: Controlling how data are displayed.

- A *value label* (possibly). This is the mechanism that allows Stata to store numerical data while displaying text. See [GSU] 9 Labeling data and [U] 12.6.3 Value labels.

- A *variable label*, which is what you call the variable when communicating with other people. Stata uses the variable label when making tables, as we will see.

A dataset is far more than simply the data it contains. It is also information that makes the data usable by someone other than the original creator.

Although describing the data tells us something about the structure of the data, it says little about the data themselves. The data can be summarized by clicking on Statistics > Summaries, tables, and tests > Summary and descriptive statistics > Summary statistics and clicking on the OK button. You could also type `summarize` in the Command window and press Enter. The result is a table containing summary statistics about all the variables in the dataset:

<table>
<thead>
<tr>
<th>Variable</th>
<th>Obs</th>
<th>Mean</th>
<th>Std. Dev.</th>
<th>Min</th>
<th>Max</th>
</tr>
</thead>
<tbody>
<tr>
<td>make</td>
<td>0</td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>price</td>
<td>74</td>
<td>6165.257</td>
<td>2949.496</td>
<td>3291</td>
<td>15906</td>
</tr>
<tr>
<td>mpg</td>
<td>74</td>
<td>21.2973</td>
<td>5.785503</td>
<td>12</td>
<td>41</td>
</tr>
<tr>
<td>rep78</td>
<td>69</td>
<td>3.405797</td>
<td>.9899323</td>
<td>1</td>
<td>5</td>
</tr>
<tr>
<td>headroom</td>
<td>74</td>
<td>2.993243</td>
<td>.8459948</td>
<td>1.5</td>
<td>5</td>
</tr>
<tr>
<td>trunk</td>
<td>74</td>
<td>13.75676</td>
<td>4.277404</td>
<td>5</td>
<td>23</td>
</tr>
<tr>
<td>weight</td>
<td>74</td>
<td>3019.459</td>
<td>777.1936</td>
<td>1760</td>
<td>4840</td>
</tr>
<tr>
<td>length</td>
<td>74</td>
<td>187.9324</td>
<td>22.26634</td>
<td>142</td>
<td>233</td>
</tr>
<tr>
<td>turn</td>
<td>74</td>
<td>39.64865</td>
<td>4.399354</td>
<td>31</td>
<td>51</td>
</tr>
<tr>
<td>displacement</td>
<td>74</td>
<td>197.2973</td>
<td>91.83722</td>
<td>79</td>
<td>425</td>
</tr>
<tr>
<td>gear_ratio</td>
<td>74</td>
<td>3.014865</td>
<td>.4562871</td>
<td>2.19</td>
<td>3.89</td>
</tr>
<tr>
<td>foreign</td>
<td>74</td>
<td>.2972873</td>
<td>.4601885</td>
<td>0</td>
<td>1</td>
</tr>
</tbody>
</table>

From this simple summary, we can learn a bit about the data. First of all, the prices are nothing like today’s car prices—of course, these cars are now antiques. We can see that the gas mileages are not particularly good. Automobile aficionados can get a feel for other esoteric characteristics.

There are two other important items here:

- The `make` variable is listed as having no observations. It really has no numerical observations because it is a string (text) variable.

- The `rep78` variable has five fewer observations than the other numerical variables. This implies that `rep78` has five missing values.

Although we could use the `summarize` and `describe` commands to get a bird’s eye view of the dataset, Stata has a command that gives a good in-depth description of the structure, contents, and values of the variables: the `codebook` command. Either type `codebook` in the Command window and press Enter or navigate the menus to Data > Describe data > Describe data contents (codebook) and click on OK. Look over the output to see that much can be learned from this simple command.
You can scroll back in the Results window to see earlier results, if need be. We will focus on the output for `make`, `rep78`, and `foreign`.

To start our investigation, we would like to run the `codebook` command on just one variable, say, `make`. We can do this, as usual, with menus or the command line. To get the `codebook` output for `make` with the menus, start by navigating to `Data > Describe data > Describe data contents (codebook)`. When the dialog appears, there are multiple ways to tell Stata to consider only the `make` variable:

- We could type `make` into the `Variables` field.
- The `Variables` field is a combo-box control that accepts variable names. Clicking on the drop triangle to the right of the `Variables` field displays a list of the variables from the current dataset. Selecting a variable from the list will, in this case, enter the variable name into the edit field.

A much easier solution is to type `codebook make` in the Command window and then press `Enter`. The result is informative:

```plaintext
  . codebook make

  make       Make and Model
  type: string (str18), but longest is str17
  unique values: 74
  missing "": 0/74
  examples:    "Cad. Deville"
               "Dodge Magnum"
               "Merc. XR-7"
               "Pont. Catalina"
  warning:    variable has embedded blanks
```

The first line of the output tells us the variable name (`make`) and the variable label (`Make and Model`). The variable is stored as a string (which is another way of saying “text”) with a maximum length of 18 characters, though a size of only 17 characters would be enough. All the values are unique, so if need be, `make` could be used as an identifier for the observations—something that is often useful when putting together data from multiple sources or when trying to weed out errors from the dataset. There are no missing values, but there are blanks within the makes. This latter fact could be useful if we were expecting `make` to be a one-word string variable.

Syntax note: Telling the `codebook` command to run on the `make` variable is an example of using a `varlist` in Stata’s syntax.

Looking at the `foreign` variable can teach us about value labels. We would like to look at the `codebook` output for this variable, and on the basis of our latest experience, it would be easy to type `codebook foreign` into the Command window (from here on, we will not explicitly say to press the `Enter` key) to get the following output:
introducing stata—sample session

. codebook foreign

<table>
<thead>
<tr>
<th>foreign</th>
<th>Car type</th>
</tr>
</thead>
<tbody>
<tr>
<td>type: numeric (byte)</td>
<td>label: origin</td>
</tr>
<tr>
<td>range: [0,1]</td>
<td>units: 1</td>
</tr>
<tr>
<td>unique values: 2</td>
<td>missing .: 0/74</td>
</tr>
<tr>
<td>tabulation: Freq. Numeric Label</td>
<td></td>
</tr>
<tr>
<td>52</td>
<td>0</td>
</tr>
<tr>
<td>22</td>
<td>1</td>
</tr>
</tbody>
</table>

We can glean that foreign is an indicator variable because its only values are 0 and 1. The variable has a value label that displays Domestic instead of 0 and Foreign instead of 1. There are two advantages of storing the data in this form:

- Storing the variable as a byte takes less memory because each observation uses 1 byte instead of the 8 bytes needed to store “Domestic”. This is important in large datasets. See [U] 12.2.2 Numeric storage types.
- As an indicator variable, it is easy to incorporate into statistical models. See [U] 26 Working with categorical data and factor variables.

Finally, we can learn a little about a poorly labeled variable with missing values by looking at the rep78 variable. Typing codebook rep78 into the Command window yields

. codebook rep78

<table>
<thead>
<tr>
<th>rep78</th>
<th>Repair Record 1978</th>
</tr>
</thead>
<tbody>
<tr>
<td>type: numeric (int)</td>
<td></td>
</tr>
<tr>
<td>range: [1,5]</td>
<td>units: 1</td>
</tr>
<tr>
<td>unique values: 5</td>
<td>missing .: 5/74</td>
</tr>
<tr>
<td>tabulation: Freq. Value</td>
<td></td>
</tr>
<tr>
<td>2</td>
<td>1</td>
</tr>
<tr>
<td>8</td>
<td>2</td>
</tr>
<tr>
<td>30</td>
<td>3</td>
</tr>
<tr>
<td>18</td>
<td>4</td>
</tr>
<tr>
<td>11</td>
<td>5</td>
</tr>
<tr>
<td>5</td>
<td>.</td>
</tr>
</tbody>
</table>

rep78 appears to be a categorical variable, but because of lack of documentation, we do not know what the numbers mean. (To see how we would label the values, see Changing data in [GSU] 6 Using the Data Editor and see [GSU] 9 Labeling data.) This variable has five missing values, meaning that there are five observations for which the repair record is not recorded. We could use the Data Editor to investigate these five observations, but we will do this by using the Command window only because doing so is much simpler.

The command equivalent to clicking on the Data Editor (Browse) button is browse. We would like to browse only those observations for which rep78 is missing, so we could type
From this, we see that the . entries are indeed missing values. The . is the default numerical missing value; Stata also allows .a, ..., .z as user missing values, but we do not have any in our dataset. See [U] 12.2.1 Missing values. Close the Data Editor after you are satisfied with this statement.

Syntax note: Using the if qualifier above is what allowed us to look at a subset of the observations.

Looking through the data lends no clues about why these particular data are missing. We decide to check the source of the data to see if the missing values were originally missing or if they were omitted in error. Listing the makes of the cars whose repair records are missing will be all we need because we saw earlier that the values of make are unique. This can be done with the menus and a dialog:

1. Select Data > Describe data > List data.
2. Click on the drop triangle to the right of the Variables field to show the variable names.
3. Click on make to enter it into the Variables field.
4. Click on the by/if/in tab in the dialog.
5. Type missing(rep78) into the If: (expression) box.
6. Click on Submit. Stata executes the proper command but the dialog remains open. Submit is useful when experimenting, exploring, or building complex commands. We will primarily use Submit in the examples. You may click on OK in its place if you like, and it will close the dialog box.

The same ends could be achieved by typing list make if missing(rep78) in the Command window. The latter is easier once you know that the command list is used for listing observations. In any case, here is the output:

```stata
data: missing(rep78)

<table>
<thead>
<tr>
<th>make</th>
<th>price</th>
<th>mpg</th>
<th>rep78</th>
<th>headroom</th>
<th>trunk</th>
</tr>
</thead>
<tbody>
<tr>
<td>AMC Spirit</td>
<td>3,799</td>
<td>22</td>
<td>.</td>
<td>3.8</td>
<td>12</td>
</tr>
<tr>
<td>Buick Opel</td>
<td>4,453</td>
<td>26</td>
<td>.</td>
<td>3.8</td>
<td>10</td>
</tr>
<tr>
<td>Plym. Sapporo</td>
<td>6,486</td>
<td>26</td>
<td>.</td>
<td>1.5</td>
<td>8</td>
</tr>
<tr>
<td>Pont. Phoenix</td>
<td>4,424</td>
<td>19</td>
<td>.</td>
<td>3.5</td>
<td>13</td>
</tr>
<tr>
<td>Peugeot 604</td>
<td>12,990</td>
<td>14</td>
<td>.</td>
<td>3.5</td>
<td>14</td>
</tr>
</tbody>
</table>
```
At this point we should find the original reference to see if the data were truly missing or if they could be resurrected. See \textit{GSU} 10 \textbf{Listing data and basic command syntax} for more information about all that can be done with the \texttt{list} command.

Syntax note: This command uses two new concepts for Stata commands—the \texttt{if} qualifier and the \texttt{missing()} function. The \texttt{if} qualifier restricts the observations on which the command runs to only those observations for which the expression is true. See \textit{U} 11.1.3 \texttt{if exp}. The \texttt{missing()} function tests each observation to see if it contains a missing value. See \textit{FN} Programming functions.

Now that we have a good idea about the underlying dataset, we can investigate the data themselves.

\section*{Descriptive statistics}

We saw above that the \texttt{summarize} command gave brief summary statistics about all the variables. Suppose now that we became interested in the prices while summarizing the data because they seemed fantastically low (it was 1978, after all). To get an in-depth look at the \texttt{price} variable, we can use the menus and a dialog:

1. Select \texttt{Statistics > Summaries, tables, and tests > Summary and descriptive statistics > Summary statistics}.
2. Enter or select \texttt{price} in the \texttt{Variables} field.
3. Select \texttt{Display additional statistics}.
4. Click on \texttt{Submit}.

Syntax note: As can be seen from the Results window, typing \texttt{summarize price, detail} will get the same result. The portion after the comma contains \textit{options} for Stata commands; hence, \texttt{detail} is an example of an option.
. summarize price, detail

<table>
<thead>
<tr>
<th>Percentiles</th>
<th>Smallest</th>
</tr>
</thead>
<tbody>
<tr>
<td>1%</td>
<td>3291</td>
</tr>
<tr>
<td>5%</td>
<td>3748</td>
</tr>
<tr>
<td>10%</td>
<td>3895</td>
</tr>
<tr>
<td>25%</td>
<td>4195</td>
</tr>
<tr>
<td>50%</td>
<td>5006.5</td>
</tr>
<tr>
<td>75%</td>
<td>6342</td>
</tr>
<tr>
<td>90%</td>
<td>11385</td>
</tr>
<tr>
<td>95%</td>
<td>13466</td>
</tr>
<tr>
<td>99%</td>
<td>15906</td>
</tr>
</tbody>
</table>

<table>
<thead>
<tr>
<th>Largest</th>
</tr>
</thead>
<tbody>
<tr>
<td>Mean</td>
</tr>
<tr>
<td>Std. Dev.</td>
</tr>
<tr>
<td>Variance</td>
</tr>
<tr>
<td>Skewness</td>
</tr>
<tr>
<td>Kurtosis</td>
</tr>
</tbody>
</table>

From the output, we can see that the median price of the cars in the dataset is only $5,006.50! We can also see that the four most expensive cars are all priced between $13,400 and $16,000. If we wished to browse the most expensive cars (and gain some experience with features of the Data Editor), we could start by clicking on the Data Editor (Browse) button.

Once the Data Editor is open, we can click on the Filter observations... button, to bring up the Filter observations dialog. We can look at the expensive cars by putting `price > 13000` in the Filter by expression field:

Pressing the Apply filter button filters the data, and we can see that the expensive cars are two Cadillacs and two Lincolns, which were not designed for gas mileage:
We now decide to turn our attention to foreign cars and repairs because as we glanced through the data, it appeared that the foreign cars had better repair records. (We do not know exactly what the categories 1, 2, 3, 4, and 5 mean, but we know the Chevy Monza was known for breaking down.) Let’s start by looking at the proportion of foreign cars in the dataset along with the proportion of cars with each type of repair record. We can do this with one-way tables. The table for foreign cars can be done with menus and a dialog starting with Statistics > Summaries, tables, and tests > Frequency tables > One-way table and then choosing the variable foreign in the Categorical variable field. Clicking on Submit yields

```
. tabulate foreign

<table>
<thead>
<tr>
<th>Car type</th>
<th>Freq.</th>
<th>Percent</th>
<th>Cum.</th>
</tr>
</thead>
<tbody>
<tr>
<td>Domestic</td>
<td>52</td>
<td>70.27</td>
<td>70.27</td>
</tr>
<tr>
<td>Foreign</td>
<td>22</td>
<td>29.73</td>
<td>100.00</td>
</tr>
<tr>
<td>Total</td>
<td>74</td>
<td>100.00</td>
<td></td>
</tr>
</tbody>
</table>
```

We see that roughly 70% of the cars in the dataset are domestic, whereas 30% are foreign. The value labels are used to make the table so that the output is nicely readable.

Syntax note: We also see that this one-way table could be made by using the `tabulate` command together with one variable, foreign. Making a one-way table for the repair records is simple—it will be simpler if done with the Command window. Typing `tabulate rep78` yields

```
. tabulate rep78

<table>
<thead>
<tr>
<th></th>
<th></th>
<th></th>
<th></th>
</tr>
</thead>
<tbody>
<tr>
<td>1</td>
<td>2</td>
<td>2.90</td>
<td>2.90</td>
</tr>
<tr>
<td>2</td>
<td>8</td>
<td>11.59</td>
<td>14.49</td>
</tr>
<tr>
<td>3</td>
<td>30</td>
<td>43.48</td>
<td>57.97</td>
</tr>
<tr>
<td>4</td>
<td>18</td>
<td>26.09</td>
<td>84.06</td>
</tr>
<tr>
<td>5</td>
<td>11</td>
<td>15.94</td>
<td>100.00</td>
</tr>
<tr>
<td>Total</td>
<td>69</td>
<td>100.00</td>
<td></td>
</tr>
</tbody>
</table>
```
We can see that most cars have repair records of 3 and above, though the lack of value labels makes us unsure what a “3” means. Take our word for it that 1 means a poor repair record and 5 means a good repair record. The five missing values are indirectly evident because the total number of observations listed is 69 rather than 74.

These two one-way tables do not help us compare the repair records of foreign and domestic cars. A two-way table would help greatly, which we can get by using the menus and a dialog:

1. Select **Statistics > Summaries, tables, and tests > Frequency tables > Two-way table with measures of association**.
2. Choose rep78 as the **Row variable**.
3. Choose foreign as the **Column variable**.
4. It would be nice to have the percentages within the foreign variable, so check the **Within-row relative frequencies** checkbox.
5. Click on **Submit**.

Here is the resulting output:

```
. tabulate rep78 foreign, row

<table>
<thead>
<tr>
<th>Repair Record</th>
<th>Car type</th>
<th></th>
<th></th>
</tr>
</thead>
<tbody>
<tr>
<td></td>
<td>Domestic</td>
<td>Foreign</td>
<td>Total</td>
</tr>
<tr>
<td>1978</td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>1</td>
<td>2</td>
<td>0</td>
<td>2</td>
</tr>
<tr>
<td></td>
<td>100.00</td>
<td>0.00</td>
<td>100.00</td>
</tr>
<tr>
<td>2</td>
<td>8</td>
<td>0</td>
<td>8</td>
</tr>
<tr>
<td></td>
<td>100.00</td>
<td>0.00</td>
<td>100.00</td>
</tr>
<tr>
<td>3</td>
<td>27</td>
<td>3</td>
<td>30</td>
</tr>
<tr>
<td></td>
<td>90.00</td>
<td>10.00</td>
<td>100.00</td>
</tr>
<tr>
<td>4</td>
<td>9</td>
<td>9</td>
<td>18</td>
</tr>
<tr>
<td></td>
<td>50.00</td>
<td>50.00</td>
<td>100.00</td>
</tr>
<tr>
<td>5</td>
<td>2</td>
<td>9</td>
<td>11</td>
</tr>
<tr>
<td></td>
<td>18.18</td>
<td>81.82</td>
<td>100.00</td>
</tr>
<tr>
<td>Total</td>
<td>48</td>
<td>21</td>
<td>69</td>
</tr>
<tr>
<td></td>
<td>69.57</td>
<td>30.43</td>
<td>100.00</td>
</tr>
</tbody>
</table>
```

The output indicates that foreign cars are generally much better than domestic cars when it comes to repairs. If you like, you could repeat the previous dialog and try some of the hypothesis tests available from the dialog. We will abstain.

Syntax note: We see that typing the command `tabulate rep78 foreign, row` would have given us the same table. Thus using `tabulate` with two variables yields a two-way table. It makes sense that `row` is an option—we went out of our way to check it in the dialog. Using the `row` option allows us to change the behavior of the `tabulate` command from its default.

Continuing our exploratory tour of the data, we would like to compare gas mileages between foreign and domestic cars, starting by looking at the summary statistics for each group by itself. A
direct way to do this would be to use `if` qualifiers to summarize `mpg` for each of the two values of `foreign` separately:

```
. summarize mpg if foreign==0
Variable | Obs  | Mean  | Std. Dev.   | Min  | Max  |
---------|------|-------|-------------|------|------|
mpg      | 52   | 19.82 | 4.743297    | 12   | 34   |
```

```
. summarize mpg if foreign==1
Variable | Obs  | Mean  | Std. Dev.   | Min  | Max  |
---------|------|-------|-------------|------|------|
mpg      | 22   | 24.77 | 6.611187    | 14   | 41   |
```

It appears that foreign cars get somewhat better gas mileage—we will test this soon.

Syntax note: We needed to use a double equal sign (==) for testing equality. The double equal sign could be familiar to you if you have programmed before. If it is unfamiliar, be aware that it is a common source of errors when initially using Stata. Thinking of equality as “exactly equal” can cut down on typing errors.

There are two other methods that we could have used to produce these summary statistics. These methods are worth knowing because they are less error-prone. The first method duplicates the concept of what we just did by exploiting Stata’s ability to run a command on each of a series of nonoverlapping subsets of the dataset. To use the menus and a dialog, do the following:

1. Select Statistics > Summaries, tables, and tests > Summary and descriptive statistics > Summary statistics and click on the Reset button.
2. Select `mpg` in the Variables field.
3. Select the Standard display option (if it is not already selected).
4. Click on the by/if/in tab.
5. Check the Repeat command by groups checkbox.
6. Select or type `foreign` in the Variables that define groups field.
7. Submit the command.

You can see that the results match those from above. They have a better appearance than the two commands above because the value labels Domestic and Foreign are used rather than the numerical values. The method is more appealing because the results were produced without needing to know the possible values of the grouping variable ahead of time.

```
. by foreign, sort: summarize mpg
```

```
-> foreign = Domestic
Variable | Obs  | Mean  | Std. Dev.   | Min  | Max  |
---------|------|-------|-------------|------|------|
mpg      | 52   | 19.82 | 4.743297    | 12   | 34   |
-> foreign = Foreign
Variable | Obs  | Mean  | Std. Dev.   | Min  | Max  |
---------|------|-------|-------------|------|------|
mpg      | 22   | 24.77 | 6.611187    | 14   | 41   |
```

Syntax note: There is something different about the equivalent command that appears above: it contains a prefix command called a by prefix. The by prefix has its own option, namely, sort, to ensure that like members are adjacent to each other before being summarized. The by prefix command is
important for understanding data manipulation and working with subpopulations within Stata. Make good note of this example, and consult [U] 11.1.2 by varlist: and [U] 13.7 Explicit subscripting for more information. Stata has other prefix commands for specialized treatment of commands, as explained in [U] 11.1.10 Prefix commands.

The third method for tabulating the differences in gas mileage across the cars’ origins involves thinking about the structure of desired output. We need a one-way table of automobile types (foreign versus domestic) within which we see information about gas mileages. Looking through the menus yields the menu item Statistics > Summaries, tables, and tests > Other tables > Table of means, std. dev., and frequencies. Selecting this, entering foreign for Variable 1 and mpg for the Summarize variable, and submitting the command yields a nice table:

### . tabulate foreign, summarize(mpg)

<table>
<thead>
<tr>
<th>Car type</th>
<th>Summary of Mileage (mpg)</th>
<th>Freq.</th>
</tr>
</thead>
<tbody>
<tr>
<td></td>
<td>Mean</td>
<td>Std. Dev.</td>
</tr>
<tr>
<td>Domestic</td>
<td>19.826923</td>
<td>4.7432972</td>
</tr>
<tr>
<td>Foreign</td>
<td>24.772727</td>
<td>6.6111869</td>
</tr>
<tr>
<td>Total</td>
<td>21.297297</td>
<td>5.7855032</td>
</tr>
</tbody>
</table>

The equivalent command is evidently `tabulate foreign, summarize(mpg)`.

Syntax note: This is a one-way table, so `tabulate` uses one variable. The variable being summarized is passed to the `tabulate` command with an option. Though we will not do it here, the `summarize()` option can also be used with two-way tables.

### A simple hypothesis test

We would like to run a hypothesis test for the difference in the mean gas mileages. Under the menus, Statistics > Summaries, tables, and tests > Classical tests of hypotheses > t test (mean-comparison test) leads to the proper dialog. Select the Two-sample using groups radio button, enter `mpg` for the Variable name and `foreign` for the Group variable name, and `Submit` the dialog. The results are

```stata
. ttest mpg, by(foreign)
Two-sample t test with equal variances

<table>
<thead>
<tr>
<th></th>
<th>Obs</th>
<th>Mean</th>
<th>Std. Err.</th>
<th>Std. Dev.</th>
<th>[95% Conf. Interval]</th>
</tr>
</thead>
<tbody>
<tr>
<td>Domestic</td>
<td>52</td>
<td>19.826923</td>
<td>.657777</td>
<td>4.743297</td>
<td>18.50638  21.14747</td>
</tr>
<tr>
<td>Foreign</td>
<td>22</td>
<td>24.772727</td>
<td>1.40951</td>
<td>6.611187</td>
<td>21.84149  27.70396</td>
</tr>
<tr>
<td>combined</td>
<td>74</td>
<td>21.297297</td>
<td>.6726511</td>
<td>5.785503</td>
<td>19.9569   22.63769</td>
</tr>
<tr>
<td>diff</td>
<td>-4.945804</td>
<td>1.362162</td>
<td>-7.661225</td>
<td>2.230384</td>
<td></td>
</tr>
</tbody>
</table>
```

From this, we could conclude that the mean gas mileage for foreign cars is different from that of domestic cars (though we really ought to have wanted to test this before snooping through the data).
We can also conclude that the command, `ttest mpg, by(foreign)`, is easy enough to remember. Feel free to experiment with unequal variances, various approximations to the number of degrees of freedom, and the like.

Syntax note: The `by()` option used here is not the same as the `by` prefix command used earlier. Although it has a similar conceptual meaning, its usage is different because it is a particular option for the `ttest` command.

**Descriptive statistics—correlation matrices**

We now change our focus from exploring categorical relationships to exploring numerical relationships: we would like to know if there is a correlation between miles per gallon and weight. We select `Statistics > Summaries, tables, and tests > Summary and descriptive statistics > Correlations and covariances` in the menus. Entering `mpg` and `weight`, either by clicking or by typing, and then submitting the command yields

```
. correlate mpg weight
(obs=74)

          mpg       weight
mpg   1.0000
weight -0.8072   1.0000
```

The equivalent command for this is natural: `correlate mpg weight`. There is a negative correlation, which is not surprising because heavier cars should be harder to push about.

We could see how the correlation compares for foreign and domestic cars by using our knowledge of the `by` prefix. We can reuse the `correlate` dialog or use the menus as before if the dialog is closed. Click on the `by/if/in` tab, check the `Repeat command by groups` checkbox, and enter the `foreign` variable to define the groups. As done on page ‘bywsort’, a simple `by foreign, sort:` prefix in front of our previous command would work, too:

```
. by foreign, sort: correlate mpg weight

-> foreign = Domestic
(obs=52)

          mpg       weight
mpg   1.0000
weight -0.8759   1.0000

-> foreign = Foreign
(obs=22)

          mpg       weight
mpg   1.0000
weight -0.6829   1.0000
```

We see from this that the correlation is not as strong among the foreign cars.

Syntax note: Although we used the `correlate` command to look at the correlation of two variables, Stata can make correlation matrices for an arbitrary number of variables:
Graphing data

We are about to make some graphs. You must be using Stata(GUI) to see them.

We have found several things in our investigations so far: We know that the average MPG of domestic and foreign cars differs. We have learned that domestic and foreign cars differ in other ways as well, such as in frequency-of-repair record. We found a negative correlation between MPG and weight—as we would expect—but the correlation appears stronger for domestic cars.

We would now like to examine, with an eye toward modeling, the relationship between MPG and weight, starting with a graph. We can start with a scatterplot of mpg against weight. The command for this is simple: scatter mpg weight. Using the menus requires a few steps because the graphs in Stata may be customized heavily.

1. Select Graphics > Twoway graph (scatter, line, etc.).
2. Click on the Create... button.
3. Select the Basic plots radio button (if it is not already selected).
4. Select Scatter as the basic plot type (if it is not already selected).
5. Select mpg as the Y variable and weight as the X variable.
6. Click on the Submit button.

The Results window shows the command that was issued from the menu:

```
.scatter mpg weight
```

The command issued when the dialog was submitted is a bit more complex than the command suggested above. There is good reason for this: the more complex structure allows combining and overlaying graphs, as we will soon see. In any case, the graph that appears is
We see the negative correlation in the graph, though the relationship appears to be nonlinear.

Note: When you draw a graph, the Graph window appears, probably covering up your Results window. Click on the main Stata window to get the Results window back on top. Want to see the graph again? Click on the Graph button. See The Graph button in [GSU] 14 Graphing data for more information about the Graph button.

Note: You must be using X Windows and Stata(GUI) for the graph to appear. If you are using Stata(console), a graph will not appear when you use the scatter command. For more information on obtaining graphs when using Stata(console), see the [G] Graphics Reference Manual.

We would now like to see how the different correlations for foreign and domestic cars are manifested in scatterplots. It would be nice to see a scatterplot for each type of car, along with a scatterplot for all the data.

Syntax note: Because we are looking at subgroups, this looks as if it is a job for the by prefix. Let’s see if this is what we really should use.

Start as before:
1. Select Graphics > Twoway graph (scatter, line, etc.) from the menus.
2. If the Plot 1 dialog is still visible, click on the Accept button and skip to step 4.
3. Go through the process on the previous page to create the graph.
4. Click on the By tab of the twoway - Twoway graphs dialog.
5. Check the Draw subgraphs for unique values of variables checkbox.
6. Enter foreign in the Variables field.
7. Check the Add a graph with totals checkbox.
8. Click on the Submit button.
The command and the associated graph are

```
. twoway (scatter mpg weight), by(foreign, total)
```

The graphs show that the relationship is nonlinear for both types of cars.

Syntax note: To make the graphs for the combined subgroups, we ended up using a by() option, not a by prefix. If we had used a by prefix, separate graphs would have been generated instead of the combined graph created by the by() option.

**Model fitting: Linear regression**

After looking at the graphs, we would like to fit a regression model that predicts MPG from the weight and type of the car. From the graphs, we see that the relationship is nonlinear, so we will try modeling MPG as a quadratic in weight. Also from the graphs, we judge the relationship to be different for domestic and foreign cars. We will include an indicator (dummy) variable for foreign and evaluate afterward whether this adequately describes the difference. Thus we will fit the model

\[
mpg = \beta_0 + \beta_1 \text{weight} + \beta_2 \text{weight}^2 + \beta_3 \text{foreign} + \epsilon
\]

foreign is already an indicator (0/1) variable, but we need to create the weight-squared variable. This can be done with the menus, but here using the command line is simpler. Type

```
. generate wtsq = weight^2
```
Now that we have all the variables we need, we can run a linear regression. We will use the menus and see that the command is also simple. To use the menus, select **Statistics > Linear models and related > Linear regression**. In the resulting dialog, choose `mpg` as the *Dependent variable* and `weight`, `wtsq`, and `foreign` as the *Independent variables*. **Submit** the command. Here is the equivalent simple `regress` command and the resulting analysis-of-variance table.

```
. regress mpg weight wtsq foreign
```

<table>
<thead>
<tr>
<th>Source</th>
<th>SS</th>
<th>df</th>
<th>MS</th>
<th>Number of obs = 74</th>
</tr>
</thead>
<tbody>
<tr>
<td>Model</td>
<td>1689.15372</td>
<td>3</td>
<td>563.05124</td>
<td>F(3, 70) = 52.25</td>
</tr>
<tr>
<td>Residual</td>
<td>754.30574</td>
<td>70</td>
<td>10.7757963</td>
<td>Prob &gt; F = 0.0000</td>
</tr>
<tr>
<td>Total</td>
<td>2443.45946</td>
<td>73</td>
<td>33.4720474</td>
<td>R-squared = 0.6913</td>
</tr>
</tbody>
</table>

Adj R-squared = 0.6781

Root MSE = 3.2827

| mpg     | Coef.   | Std. Err. | t     | P>|t|  | [95% Conf. Interval]       |
|---------|---------|-----------|-------|-----|-----------------------------|
| weight  | -0.0165729 | 0.0039692 | -4.18 | 0.000 | -0.0244892 - 0.0086567    |
| wtsq    | 1.59e-06  | 6.25e-07  | 2.55  | 0.013 | 3.45e-07 - 2.84e-06       |
| foreign | -2.2035  | 1.059246  | -2.08 | 0.041 | -4.3161 - 0.0909002      |
| _cons   | 56.53884 | 6.197383  | 9.12  | 0.000 | 44.17855 - 68.89913      |

The results look encouraging, so we will plot the predicted values on top of the scatterplots for each of the types of cars. To do this, we need the predicted, or fitted, values. This can be done with the menus, but doing it in the Command window is simple enough. We will create a new variable, `mpghat`, to hold the predicted MPG for each car. Type

```
. predict mpghat
```

The output from this command is simply a notification. Go over to the Variables window and scroll to the bottom to confirm that there is now an `mpghat` variable. If you were to try this command when `mpghat` already existed, Stata would refuse to overwrite your data:

```
. predict mpghat
         variable mpghat already defined
          r(110);           
```

The `predict` command, when used after a regression, is called a *postestimation command*. As specified, it creates a new variable called `mpghat` equal to

\[ -0.0165729 \text{weight} + 1.59 \times 10^{-6} \text{wtsq} - 2.2035 \text{foreign} + 56.53884 \]

For careful model fitting, there are several features available to you after estimation—one is calculating predicted values. Be sure to read [U] 20 Estimation and postestimation commands.

We can now graph the data and the predicted curve to evaluate separately the fit on the foreign and domestic data to determine if our shift parameter is adequate. We can draw both graphs together. Using the menus and a dialog, do the following:

1. Select **Graphics > Twoway graph (scatter, line, etc.)** from the menus.
2. If there are any plots listed, click on the **Reset** button, ⚪️, to clear the dialog box.
3. Create the graph for mpg versus weight:
   a. Click on the Create... button.
   b. Be sure that Basic plots and Scatter are selected.
   c. Select mpg as the Y variable and weight as the X variable.
   d. Click on Accept.

4. Create the graph showing mpghat versus weight
   a. Click on the Create... button.
   b. Select Basic plots and Line.
   c. Select mpghat as the Y variable and weight as the X variable.
   d. Check the Sort on x variable box. Doing so ensures that the lines connect from smallest to largest weight values, instead of the order in which the data happen to be.
   e. Click on Accept.

5. Show two plots, one each for domestic and foreign cars, on the same graph:
   a. Click on the By tab.
   b. Check the Draw subgraphs for unique values of variables checkbox.
   c. Enter foreign in the Variables field.

6. Click on the Submit button.

Here are the resulting command and graph:

```
. twoway (scatter mpg weight) (line mpghat weight, sort), by(foreign)
```

![Graphs by Car type](image-url)
Here we can see the reason for enclosing the separate `scatter` and `line` commands in parentheses: they can thereby be overlaid by submitting them together. The fit of the plots looks good and is cause for initial excitement. So much excitement, in fact, that we decide to print the graph and show it to an engineering friend. We print the graph, being careful to print the graph (and not all our results), by choosing **File > Print...** from the Graph window menu bar.

When we show our graph to our engineering friend, she is concerned. “No,” she says. “It should take twice as much energy to move 2,000 pounds 1 mile compared with moving 1,000 pounds the same distance: therefore, it should consume twice as much gasoline. Miles per gallon is not quadratic in weight; gallons per mile is a linear function of weight. Don’t you remember any physics?”

We try out what she says. We need to generate an energy-per-distance variable and make a scatterplot. Here are the commands that we would need—note their similarity to commands issued earlier in the session. There is one new command, the `label variable` command, which allows us to give the `gpm100m` variable a **variable label** so that the graph is labeled nicely.

```
. generate gp100m = 100/mpg
. label variable gp100m "Gallons per 100 miles"
. twoway (scatter gp100m weight), by(foreign, total)
```

![Graphs by Car type](image-url)
Introducing Stata—sample session

Sadly satisfied that the engineer is indeed correct, we rerun the regression:

\[
\text{. regress gp100m weight foreign}
\]

<table>
<thead>
<tr>
<th>Source</th>
<th>SS</th>
<th>df</th>
<th>MS</th>
<th>Number of obs = 74</th>
</tr>
</thead>
<tbody>
<tr>
<td>Model</td>
<td>91.1761694</td>
<td>2</td>
<td>45.5880847</td>
<td>F(2, 71) = 113.97</td>
</tr>
<tr>
<td>Residual</td>
<td>28.4000913</td>
<td>71</td>
<td>.400001287</td>
<td>Prob &gt; F = 0.0000</td>
</tr>
<tr>
<td>Total</td>
<td>119.576261</td>
<td>73</td>
<td>1.63803097</td>
<td>R-squared = 0.7625</td>
</tr>
</tbody>
</table>

Root MSE = 0.63246

| gp100m   | Coef.     | Std. Err. | t     | P>|t|     | [95% Conf. Interval] |
|----------|-----------|-----------|-------|---------|---------------------|
| weight   | .0016254  | .0001183  | 13.74 | 0.000   | .0013896 .0018612  |
| foreign  | .6220535  | .1997381  | 3.11  | 0.003   | .2237871 1.02032  |
| _cons    | -.0734839 | .4019932  | -0.18 | 0.855   | -.8750354 .7280677 |

We find that foreign cars had better gas mileage than domestic cars in 1978 because they were so light. According to our model, a foreign car with the same weight as a domestic car would use an additional 5/8 gallon (or 5 pints) of gasoline per 100 miles driven. With this conclusion, we are satisfied with our analysis.

Commands versus menus

In this chapter, you have seen that Stata can operate either with menu choices and dialogs or with the Command window. As you become more familiar with Stata, you will find that the Command window is typically much faster for oft-used commands, whereas the menus and dialogs are faster when building up complex commands, such as those that create graphs.

One of Stata’s great strengths is the consistency of its command syntax. Most of Stata’s commands share the following syntax, where square brackets mean that something is optional, and a varlist is a list of variables.

\[
[\text{prefix:}] \text{command} [\text{varlist}] [\text{if}] [\text{in}] [\text{weight}] [ , \text{options}]
\]

Some general rules:
- Most commands accept prefix commands that modify their behavior; see [U] 11.1.10 Prefix commands for details. One of the common prefix commands is by.
- If an optional varlist is not specified, all the variables are used.
- if and in restrict the observations on which the command is run.
- options modify what the command does.
- Each command’s syntax is found in the system help and the reference manuals or, for commands specific to Stata for Unix, in [GSU] C Stata manual pages for Unix.
- Stata’s command syntax includes more than we have shown you here, but this introduction should get you started. For more information, see [U] 11 Language syntax and help language.

We saw examples using all the pieces of this except for the in qualifier and the weight clause. The syntax for all commands can be found in the system help along with examples—see [GSU] 4 Getting help for more information. The consistent syntax makes it straightforward to learn new commands and to read others’ commands when examining an analysis.
Here is an example of reading the syntax diagram that uses the `summarize` command from earlier in this chapter. The syntax diagram for `summarize` is typical:

```
summarize [varlist] [if] [in] [weight] [, options]
```

This means that

- `command` by itself is valid:
  - `summarize`

- `command` followed by a `varlist` (variable list) is valid:
  - `summarize mpg`
  - `summarize mpg weight`

- `command` with `if` (with or without a `varlist`) is valid:
  - `summarize if mpg>20`
  - `summarize mpg weight if mpg>20`

and so on.

You can learn about `summarize` in [R] `summarize`, or select Help > Stata command... and enter `summarize`, or type `help summarize` in the Command window.

### Keeping track of your work

It would have been useful if we had made a log of what we did so that we could conveniently look back at interesting results or track any changes that were made. You will learn to do this in [GSU] 16 Saving and printing results by using logs. Your logs will contain commands and their output—another reason to learn command syntax is so that you can remember what you have done.

To make a log file that keeps track of everything appearing in the Results window, click on the Log button, which looks like a lab notebook. Choose a place to store your log file, and give it a name, just as you would for any other document. The log file will save everything that appears in the Results window from the time you start a log file to the time you close it.

### Video example

What’s it like—Getting started in Stata

### Conclusion

This chapter introduced you to Stata’s capabilities. You should now read and work through the rest of this manual. Once you are done here, you can read the User’s Guide.